Soluzioni del compito di

Metodi Matematici e Calcolo per Ingegneria

del 9 gennaio 2006

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> t*Heaviside(1-t)
          +Heaviside(t-1)*Heaviside(2-t)
          +(3-t)*Heaviside(t-2)*Heaviside(3-t) ;
```

$$f := t \mapsto t\,\theta(1-t) + \theta(t-1)\,\theta(2-t) + (3-t)\,\theta(t-2)\,\theta(3-t)$$

```
> plot(f(t),t=0..4,axesfont=[helvetica,24]);
```



Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(t/3) ;
  f2 := t -> f(t/2)*exp(-2*t) ;
  f3 := unapply( diff(f(t),t), t) ;
```

$$f1 := t \mapsto f\left(\frac{t}{3}\right)$$

$$f2 := t \mapsto f\left(\frac{t}{2}\right) e^{-2\,t}$$

$$f3 := t \mapsto \theta(1-t) - t\,\delta(t-1) + \delta(t-1)\,\theta(2-t) - \theta(t-1)\,\delta(t-2) - \theta(t-2)\,\theta(3-t) + (3-t)\,\delta(t-2)\,\theta(3-t) - (3-t)\,\theta(t-2)\,\delta(t-3)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{1 - e^{-s} - e^{-2\,s} + e^{-3\,s}}{s^2}$$

$$\frac{1-e^{-3s}-e^{-6s}+e^{-9s}}{3s^2}$$

$$\frac{1-e^{-2s-4}-e^{-4s-8}+e^{-6s-12}}{2(s+2)^2}$$

$$\frac{1-e^{-s}-e^{-2s}+e^{-3s}}{s}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
> src := t -> t*Heaviside(1-t)+Heaviside(t-1) ;
```

$$src := t \mapsto t\,\theta(1-t)+\theta(t-1)$$

```
> plot(src(t),t=0..4,axesfont=[helvetica,24]);
```



Data la seguente equazione differenziale

```
> ode := diff(y(x),x)=src(x) ;
```

$$ode := y'(x) = x\,\theta(1-x)+\theta(x-1)$$

Con dato iniziale

```
> y0 := 10 ;
```

$$y0 := 10$$

Calcolare le soluzioni con le trasformate di Laplace.
Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s\,laplace(y(x),x,s) - y(0) = \frac{1-e^{-s}}{s^2}$$

```
> subs(laplace(y(x),x,s)=y(s),sode) ;
```

$$s\,y(s) - y(0) = \frac{1-e^{-s}}{s^2}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x),x,s) = \frac{\frac{1-e^{-s}}{s^2} + y(0)}{s}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := expand(subs( y(0)=y0,rhs(lode)) ;
```

$$ly := \frac{1}{s^3} - \frac{1}{s^3 e^s} + \frac{10}{s}$$

Espansione in fratti semplici

```
> convert(ly, fullparfrac, s);
```

$$\frac{1}{s^3} - \frac{1}{s^3 e^s} + \frac{10}{s}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := \frac{t^2}{2} - \frac{\theta(t-1)\,(t-1)^2}{2} + 10$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> yp, zp, wp := diff(y(x),x),diff(z(x),x),diff(w(x),x);
```

$$yp, zp, wp := y'(x), z'(x), w'(x)$$

```
> ode1 := 3*yp   -zp   -wp = 0 ;
  ode2 :=  -yp +3*zp   -wp = 0 ;
  ode3 :=  -yp   -zp +3*wp = exp(-x) ;
```

$$ode1 := 3\,y'(x) - z'(x) - w'(x) = 0$$

$$ode2 := -y'(x) + 3\,z'(x) - w'(x) = 0$$

$$ode3 := -y'(x) - z'(x) + 3\, w'(x) = e^{-x}$$

Con dato iniziale

```
> y0, z0, w0 := 3, 2, 1 ;
```

$$y0, z0, w0 := 3, 2, 1$$

Calcolare le soluzioni con le trasformate di Laplace.
Trasformo le equazioni differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
  sode3 := laplace(ode3,x,s) ;
```

$$sode1 := 3\, s\, laplace(y(x), x, s) - 3\, y(0) - s\, laplace(z(x), x, s) + z(0) - s\, laplace(w(x), x, s) + w(0) = 0$$

$$sode2 := -s\, laplace(y(x), x, s) + y(0) + 3\, s\, laplace(z(x), x, s) - 3\, z(0) - s\, laplace(w(x), x, s) + w(0) = 0$$

$$sode3 := -s\, laplace(y(x), x, s) + y(0) - s\, laplace(z(x), x, s) + z(0) + 3\, s\, laplace(w(x), x, s) - 3\, w(0) = \frac{1}{1+s}$$

```
> subs(laplace(y(x),x,s)=y(s),
       laplace(z(x),x,s)=z(s),
       laplace(w(x),x,s)=w(s),
       sode1);
  subs(laplace(y(x),x,s)=y(s),
       laplace(z(x),x,s)=z(s),
       laplace(w(x),x,s)=w(s),
       sode2);
  subs(laplace(y(x),x,s)=y(s),
       laplace(z(x),x,s)=z(s),
       laplace(w(x),x,s)=w(s),
       sode3);
```

$$3\, s\, y(s) - 3\, y(0) - s\, z(s) + z(0) - s\, w(s) + w(0) = 0$$

$$-s\, y(s) + y(0) + 3\, s\, z(s) - 3\, z(0) - s\, w(s) + w(0) = 0$$

$$-s\, y(s) + y(0) - s\, z(s) + z(0) + 3\, s\, w(s) - 3\, w(0) = \frac{1}{1+s}$$

Risolvo la equazione per y(s), z(s)

```
> ys,zs,ws := laplace(y(x),x,s),laplace(z(x),x,s),laplace(w(x),x,
  s);
```

$$ys, zs, ws := laplace(y(x), x, s), laplace(z(x), x, s), laplace(w(x), x, s)$$

```
> RES := solve({sode1,sode2,sode3},{ys,zs,ws});
```

$$RES := \left\{ laplace(w(x), x, s) = \frac{1 + 2\, w(0) + 2\, w(0)\, s}{2\, s\, (1+s)},\ laplace(z(x), x, s) = \frac{1 + 4\, z(0) + 4\, z(0)\, s}{4\, s\, (1+s)},\ laplace(y(x), x, s) = \frac{4\, y(0) + 4\, y(0)\, s + 1}{4\, s\, (1+s)} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL := subs(RES,y(0)=y0,z(0)=z0,w(0)=w0,<ys,zs,ws>);
```

$$SOL := \begin{bmatrix} \frac{13+12\,s}{4\,s\,(1+s)} \\ \frac{9+8\,s}{4\,s\,(1+s)} \\ \frac{3+2\,s}{2\,s\,(1+s)} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
  ww := invlaplace(SOL[3],s,x) ;
```

$$yy := \frac{13}{4} - \frac{e^{-x}}{4}$$

$$zz := \frac{9}{4} - \frac{e^{-x}}{4}$$

$$ww := \frac{3}{2} - \frac{e^{-x}}{2}$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], fullparfrac, s);
  convert(SOL[2], fullparfrac, s);
  convert(SOL[3], fullparfrac, s);
```

$$-\frac{1}{4\,(1+s)} + \frac{13}{4\,s}$$

$$-\frac{1}{4\,(1+s)} + \frac{9}{4\,s}$$

$$-\frac{1}{2\,(1+s)} + \frac{3}{2\,s}$$

Soluzione di ricorrenza con trasformata zeta

```
> restart:
```

Risolvere la seguente ricorrenza

```
> RIC := 3*f(n+2) = 2*f(n+1) + f(n) - 1 ;
```

$$RIC := 3\,f(n+2) = 2\,f(n+1) + f(n) - 1$$

Con dato iniziale

```
> INI := f(0)=0,f(1)=1;
```

$$INI := f(0) = 0, f(1) = 1$$

Usando le primitive di maple:

```
> rsolve({RIC,INI}, f(k));
```

$$-\frac{15\left(-\frac{1}{3}\right)^{k}}{16}+\frac{15}{16}-\frac{k}{4}$$

```
> simplify(%);
```

$$\frac{15\,(-1)^{k+1}\,3^{-k}}{16}+\frac{15}{16}-\frac{k}{4}$$

Usando la Z-trasformata

```
> zRIC := ztrans(RIC,n,z);
```

$$zRIC := 3\,z^2\,\mathrm{ztrans}(f(n),n,z)-3\,f(0)\,z^2-3\,f(1)\,z=2\,z\,\mathrm{ztrans}(f(n),n,z)-2\,f(0)\,z+\mathrm{ztrans}(f(n),n,z)-\frac{z}{z-1}$$

Ricavo f(z)

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,z));
```

$$zRICrhs := \mathrm{ztrans}(f(n),n,z)=\frac{3\,f(0)\,z^2+3\,f(1)\,z-2\,f(0)\,z-\frac{z}{z-1}}{3\,z^2-2\,z-1}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs);
```

$$zRICrhsINI := \mathrm{ztrans}(f(n),n,z)=\frac{3\,z-\frac{z}{z-1}}{3\,z^2-2\,z-1}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$\mathrm{ztrans}(f(n),n,z)=\frac{15}{16\,(3\,z+1)}+\frac{11}{16\,(z-1)}-\frac{1}{4\,(z-1)^2}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,z,k) ;
```

$$f(k)=-\frac{15\left(-\frac{1}{3}\right)^{k}}{16}+\frac{15}{16}-\frac{k}{4}$$

Soluzione di un sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
```

Sistema non lineare

```
> f := 2*x - y   + x*y + 1 ;
  g :=   x + 2*y - x*y - 2 ;
```

$$f := 2\,x-y+x\,y+1$$

$$g := x+2\,y-x\,y-2$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\{y=1,\ x=0\},\ \{x=2,\ y=-5\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} 2+y & -1+x \\ 1-y & -x+2 \end{bmatrix}$$

Schema di Newton

```
> Newton_update := simplify(<x,y>-J^(-1).<f,g>);
```

$$Newton\_update := -\frac{x\,(-1+y)}{3\,x-5-y}e_x+\left(\frac{3\,x\,y-y-5}{3\,x-5-y}\right)e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1}=-\frac{x_k\,(-1+y_k)}{3\,x_k-5-y_k}$$

$$y_{k+1}=\frac{3\,x_k\,y_k-y_k-5}{3\,x_k-5-y_k}$$

Tre iterate a partire da (1,2)

```
> x[0],y[0]:= 1,2 ;
```

$$x_0,\ y_0 := 1,\ 2$$

Prima iterata

```
> x[1] := evalf(subs(x=x[0],y=y[0],Newton_update[1])) ;
  y[1] := evalf(subs(x=x[0],y=y[0],Newton_update[2])) ;
```

$$x_1 := 0.2500000000$$

$$y_1 := 0.2500000000$$

Seconda iterata

```
> x[2] := evalf(subs(x=x[1],y=y[1],Newton_update[1])) ;
  y[2] := evalf(subs(x=x[1],y=y[1],Newton_update[2])) ;
```

$$x_2 := -0.04166666668$$

$$y_2 := 1.125000000$$

Terza iterata

```
> x[3] := evalf(subs(x=x[2],y=y[2],Newton_update[1])) ;
  y[3] := evalf(subs(x=x[2],y=y[2],Newton_update[2])) ;
```

$$x_3 := -0.0008333333336$$

$$y_3 := 1.002500000$$

Problema di Minimo Vincolato

```
> restart:
```

```
with(LinearAlgebra):
with(Optimization):
with(VectorCalculus):
```

Minimizzare la seguente funzione

```
> f := y;
```

$$f := y$$

Soggetta ai vicoli

```
> v := [x*y^2*z=1,x+z=1] ;
```

$$v := [x\,y^2\,z = 1,\; x+z = 1]$$

Soluzione con le primitive Maple

```
> Minimize(f, v );
```

$$[2., [x = 0.500000000000000000,\; y = 2.,\; z = 0.500000000000000000]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```

$$v1 := x\,y^2\,z - 1$$

$$v2 := x + z - 1$$

```
> g := f - lambda*v1 - mu*v2 ;
```

$$g := y - \lambda\,(x\,y^2\,z - 1) - \mu\,(x + z - 1)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```

$$F := (-\lambda\,y^2\,z - \mu)\bar{e}_x + (1 - 2\,\lambda\,x\,y\,z)\bar{e}_y + (-\lambda\,x\,y^2 - \mu)\bar{e}_z + (-x\,y^2\,z + 1)\bar{e}_\lambda + (-x - z + 1)\bar{e}_\mu$$

Soluzioni del sistema non lineare

```
> _EnvExplicit := true ;
```

$$\_EnvExplicit := true$$

```
> RES := op(sort([solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,
  mu})]));
```

$$RES := \left\{y = 2,\, z = \frac{1}{2},\, \lambda = 1,\, x = \frac{1}{2},\, \mu = -2\right\},\, \left\{z = \frac{1}{2},\, \lambda = -1,\, y = -2,\, x = \frac{1}{2},\, \mu = 2\right\}$$

Prima soluzione

```
> RES[1];
```

$$\left\{y = 2,\, z = \frac{1}{2},\, \lambda = 1,\, x = \frac{1}{2},\, \mu = -2\right\}$$

Seconda soluzione

```
> RES[2];
```

$$\left\{z = \frac{1}{2},\, \lambda = -1,\, y = -2,\, x = \frac{1}{2},\, \mu = 2\right\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
  Hv1 := Hessian(v1,[x,y,z]):
  Hv2 := Hessian(v2,[x,y,z]):
  Hf, Hv1, Hv2 ;
```

$$\left[\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{array}\right], \left[\begin{array}{ccc} 0 & 2yz & y^2 \\ 2yz & 2xz & 2xy \\ y^2 & 2xy & 0 \end{array}\right], \left[\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{array}\right]$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```

$$JH := \left[\begin{array}{ccc} y^2 z & 2xyz & xy^2 \\ 1 & 0 & 1 \end{array}\right]$$

$$NH := \left\{ \left[\begin{array}{c} -1 \\ -\frac{(-z+x)\,y}{2zx} \\ 1 \end{array}\right] \right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES[1],lambda);
  mu1     := subs(RES[1],mu);
```

$$\lambda 1 := 1$$

$$\mu 1 := -2$$

Calcolo l'Hessiano nel punto stazionario

```
> Hf1 := simplify(subs(RES[1],Hf - lambda1. Hv1 - mu1. Hv2)) ;
```

$$Hf1 := \left[\begin{array}{ccc} 0 & -2 & -4 \\ -2 & -\frac{1}{2} & -2 \\ -4 & -2 & 0 \end{array}\right]$$

L'Hessiano è indefinito!, devo controllare nello spazio dei vincoli

```
> evalf(Eigenvalues(Hf1));
```

$$\left[\begin{array}{c} 4. \\ 1.076033675 \\ -5.576033675 \end{array}\right]$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES[1],op(NH)) ;
```

$$Z1 := \left[\begin{array}{c} -1 \\ 0 \\ 1 \end{array}\right]$$

E' positivo per ogni alpha, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$8\,\alpha^2$$

```
> subs(RES[1],f);
```

$$2$$

Controllo minimo/massimo locale secondo punto

```
> lambda2 := subs(RES[2],lambda);
  mu2     := subs(RES[2],mu);
```

$$\lambda 2 := -1$$

$$\mu 2 := 2$$

```
> Hf2 := simplify(subs(RES[2],Hf - lambda2. Hv1 - mu2. Hv2)) ;
```

$$Hf2 := \begin{bmatrix} 0 & -2 & 4 \\ -2 & \frac{1}{2} & -2 \\ 4 & -2 & 0 \end{bmatrix}$$

L'Hessiano è indefinito!, devo controllare nello spazio dei vincoli

```
> evalf(Eigenvalues(Hf2));
```

$$\begin{bmatrix} -4. \\ 5.576033675 \\ -1.076033675 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z2 := subs(RES[2],op(NH)) ;
```

$$Z2 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}$$

E' negativo per ogni alpha, quindi è un massimo locale

```
> simplify(Transpose(alpha.Z2).Hf2.(alpha.Z2)) ;
```

$$-8\,\alpha^2$$

```
> subs(RES[2],f);
```

$$-2$$

Approssimazione di un problema del calcolo delle variazioni

```
> restart:
```

Integrale da minimizzare

```
> int(y(x)*sqrt(1+diff(y(x),x)^2),x=0..1);
```

$$\int_0^1 y(x)\,\sqrt{1+y'(x)^2}\,dx$$

Condizioni al contorno

```
> ya,yb := 1,1 ;
```

$$ya, yb := 1, 1$$

```
> n := 4 ;
```

$$n := 4$$

```
> h := 1/n ;
```

$$h := \frac{1}{4}$$

```
> F := sum( (y[k+1]+y[k])/2*sqrt(1+((y[k+1]-y[k])/h)^2),k=0..n-1);
```

$$F := \frac{(y_1+y_0)\sqrt{1+16 y_1^2-32 y_1 y_0+16 y_0^2}}{2} + \frac{(y_2+y_1)\sqrt{1+16 y_2^2-32 y_2 y_1+16 y_1^2}}{2} + \frac{(y_3+y_2)\sqrt{1+16 y_3^2-32 y_3 y_2+16 y_2^2}}{2} + \frac{(y_4+y_3)\sqrt{1+16 y_4^2-32 y_4 y_3+16 y_3^2}}{2}$$

```
> eqns := [seq(diff(F,y[k]),k=1..n-1),y[0]-ya,y[n]-yb]:
> vars := [seq(y[k],k=0..n)];
```

$$vars := [y_0, y_1, y_2, y_3, y_4]$$

```
> res := fsolve({op(eqns)},{op(vars)}) ;
```

$$res := \{y_0 = 1.000000000, y_4 = 1.000000000, y_1 = 0.3598476569, y_2 = 0.2134288138, y_3 = 0.3598476569\}$$

```
> yy := subs(res,vars);
  xx := seq(k/n,k=0..n);
```

$$yy := [1.000000000, 0.3598476569, 0.2134288138, 0.3598476569, 1.000000000]$$

$$xx := 0, \frac{1}{4}, \frac{1}{2}, \frac{3}{4}, 1$$

```
> plot([seq([xx[k],yy[k]],k=1..nops(yy))]);
```




```
> with(VectorCalculus):with(LinearAlgebra):
> eqns_fun := unapply(Vector([seq(simplify(subs(y[0]=ya,y[n]=yb,
  eqns[i]),sqrt,symbolic),i=1..n-1)]),
  y[1],y[2],y[3]);
  vars_reduced := [seq(y[i],i=1..n-1)];
```

$$eqns\_fun := (y\_1, y\_2, y\_3)$$

$$\mapsto \mathrm{rtable}\Bigg(1..3,$$

$$\Bigg\{1 = \frac{\sqrt{17+16\,y\_1^2-32\,y\_1}}{2} + \frac{(y\_1+1)\,(32\,y\_1-32)}{4\sqrt{17+16\,y\_1^2-32\,y\_1}}$$

$$+ \frac{\sqrt{1+16\,y\_2^2-32\,y\_2\,y\_1+16\,y\_1^2}}{2} + \frac{(y\_2+y\_1)\,(-32\,y\_2+32\,y\_1)}{4\sqrt{1+16\,y\_2^2-32\,y\_2\,y\_1+16\,y\_1^2}},$$

$$2 = \frac{\sqrt{1+16\,y\_2^2-32\,y\_2\,y\_1+16\,y\_1^2}}{2} + \frac{(y\_2+y\_1)\,(32\,y\_2-32\,y\_1)}{4\sqrt{1+16\,y\_2^2-32\,y\_2\,y\_1+16\,y\_1^2}}$$

$$+\frac{\sqrt{1+16\,y\_3^2-32\,y\_3\,y\_2+16\,y\_2^2}}{2}+\frac{(y\_3+y\_2)\,(-32\,y\_3+32\,y\_2)}{4\,\sqrt{1+16\,y\_3^2-32\,y\_3\,y\_2+16\,y\_2^2}},$$

$$3=\frac{\sqrt{1+16\,y\_3^2-32\,y\_3\,y\_2+16\,y\_2^2}}{2}+\frac{(y\_3+y\_2)\,(32\,y\_3-32\,y\_2)}{4\,\sqrt{1+16\,y\_3^2-32\,y\_3\,y\_2+16\,y\_2^2}}$$

$$+\frac{\sqrt{17-32\,y\_3+16\,y\_3^2}}{2}+\frac{(1+y\_3)\,(-32+32\,y\_3)}{4\,\sqrt{17-32\,y\_3+16\,y\_3^2}}\Bigg\},\ datatype=anything,$$

$subtype=Vector_{column}$, $storage=rectangular$, $order=Fortran\_order$,

$attributes=[coords=cartesian]\Big)$

$$vars\_reduced:=[y_1,y_2,y_3]$$

```
> J := unapply(Matrix(simplify(Jacobian(eqns_fun(x,y,z),[x,y,z]),
  sqrt,symbolic)),x,y,z) ;
```

$$J:=(x,y,z)\mapsto \mathrm{rtable}\Big(1..3,\,1..3,$$

$$\Big\{(1,1)=\frac{32\,x-32}{2\,\sqrt{17+16\,x^2-32\,x}}-\frac{(x+1)\,(32\,x-32)^2}{8\,\left(17+16\,x^2-32\,x\right)^{3/2}}+\frac{8\,(x+1)}{\sqrt{17+16\,x^2-32\,x}}$$

$$+\frac{-32\,y+32\,x}{2\,\sqrt{1+16\,y^2-32\,y\,x+16\,x^2}}-\frac{(y+x)\,(-32\,y+32\,x)^2}{8\,\left(1+16\,y^2-32\,y\,x+16\,x^2\right)^{3/2}}$$

$$+\frac{8\,(y+x)}{\sqrt{1+16\,y^2-32\,y\,x+16\,x^2}},\ (1,2)=-\frac{8\,(y+x)}{\left(1+16\,y^2-32\,y\,x+16\,x^2\right)^{3/2}},$$

$$(2,1)=-\frac{8\,(y+x)}{\left(1+16\,y^2-32\,y\,x+16\,x^2\right)^{3/2}},$$

$$(2,2)=\frac{32\,y-32\,x}{2\,\sqrt{1+16\,y^2-32\,y\,x+16\,x^2}}-\frac{(y+x)\,(32\,y-32\,x)^2}{8\,\left(1+16\,y^2-32\,y\,x+16\,x^2\right)^{3/2}}$$

$$+\frac{8\,(y+x)}{\sqrt{1+16\,y^2-32\,y\,x+16\,x^2}}$$

$$+\frac{-32\,z+32\,y}{2\,\sqrt{1+16\,z^2-32\,z\,y+16\,y^2}}-\frac{(z+y)\,(-32\,z+32\,y)^2}{8\,\left(1+16\,z^2-32\,z\,y+16\,y^2\right)^{3/2}}$$

$$+\frac{8\,(z+y)}{\sqrt{1+16\,z^2-32\,z\,y+16\,y^2}},\ (2,3)=-\frac{8\,(z+y)}{\left(1+16\,z^2-32\,z\,y+16\,y^2\right)^{3/2}},$$

$$(3,2) = -\frac{8\,(z+y)}{\left(1+16\,z^2-32\,z\,y+16\,y^2\right)^{3/2}},$$

$$(3,3) = \frac{32\,z-32\,y}{2\,\sqrt{1+16\,z^2-32\,z\,y+16\,y^2}} - \frac{(z+y)\,(32\,z-32\,y)^2}{8\,\left(1+16\,z^2-32\,z\,y+16\,y^2\right)^{3/2}}$$

$$+\frac{8\,(z+y)}{\sqrt{1+16\,z^2-32\,z\,y+16\,y^2}} + \frac{-32+32\,z}{2\,\sqrt{17-32\,z+16\,z^2}} - \frac{(1+z)\,(-32+32\,z)^2}{8\,\left(17-32\,z+16\,z^2\right)^{3/2}}$$

$$\left.+\frac{8\,(1+z)}{\sqrt{17-32\,z+16\,z^2}}\right\}, \mathit{datatype} = \mathit{anything}, \mathit{subtype} = \mathit{Matrix}, \mathit{storage} = \mathit{rectangular},$$

$$\left.\mathit{order} = \mathit{Fortran\_order}\right)$$

```
> Newton_update := Z -> Z-LinearSolve(J(Z[1],Z[2],Z[3]),eqns_fun(Z[1],Z[2],Z[3]));
```

$$\mathit{Newton\_update} := Z \rightarrow \mathit{VectorCalculus}{:-}{+}\Big(Z, \mathit{VectorCalculus}{:-}{-}\big(\mathit{LinearAlgebra}{:-}\mathit{LinearSolve}\big(J(Z_1, Z_2, Z_3), \mathit{eqns\_fun}(Z_1, Z_2, Z_3)\big)\big)\Big)$$

```
> Z0 := <1,1,1>;
```

$$Z0 := e_x + e_y + e_z$$

```
> Z1 := evalf(Newton_update(Z0));
```

$$Z1 := (0.9062500000)\,e_x + (0.8750000000)\,e_y + (0.9062500000)\,e_z$$

```
> Z2 := evalf(Newton_update(Z1));
```

$$Z2 := (0.887736987100000041)\,e_x + (0.851183777399999952)\,e_y + (0.887736987100000041)\,e_z$$

```
> Z3 := evalf(Newton_update(Z2));
```

$$Z3 := (0.886489832700000035)\,e_x + (0.849713121600000008)\,e_y + (0.886489832700000035)\,e_z$$

```
> yyy := [1,seq(Z3[k],k=1..3),1] ;
```

$$\mathit{yyy} := [1, 0.886489832700000035, 0.849713121600000008, 0.886489832700000035, 1]$$

```
> plot([seq([xx[k],yyy[k]],k=1..nops(yy))]);
```




```
> # Newton con molti piu punti!
> n := 100 ;
```

$$n := 100$$

```
> h := 1/n ;
```

$$h := \frac{1}{100}$$

```
> F := sum( (y[k+1]+y[k])/2*sqrt(1+((y[k+1]-y[k])/h)^2),k=0..n-1):
> eqns := [seq(diff(F,y[k]),k=1..n-1),y[0]-ya,y[n]-yb]:
> vars := [seq(y[k],k=0..n)];
```

$$vars := [y_0, y_1, y_2, y_3, y_4, y_5, y_6, y_7, y_8, y_9, y_{10}, y_{11}, y_{12}, y_{13}, y_{14}, y_{15}, y_{16}, y_{17}, y_{18}, y_{19}, y_{20}, y_{21}, y_{22}, y_{23}, y_{24}, y_{25}, y_{26}, y_{27}, y_{28}, y_{29}, y_{30}, y_{31}, y_{32}, y_{33}, y_{34}, y_{35}, y_{36}, y_{37}, y_{38}, y_{39}, y_{40}, y_{41}, y_{42}, y_{43}, y_{44}, y_{45}, y_{46}, y_{47}, y_{48}, y_{49}, y_{50}, y_{51}, y_{52}, y_{53}, y_{54}, y_{55}, y_{56}, y_{57}, y_{58}, y_{59}, y_{60}, y_{61}, y_{62}, y_{63}, y_{64}, y_{65}, y_{66}, y_{67}, y_{68}, y_{69}, y_{70}, y_{71}, y_{72}, y_{73}, y_{74}, y_{75}, y_{76}, y_{77}, y_{78}, y_{79}, y_{80}, y_{81}, y_{82}, y_{83}, y_{84}, y_{85}, y_{86}, y_{87}, y_{88}, y_{89}, y_{90}, y_{91}, y_{92}, y_{93}, y_{94}, y_{95}, y_{96}, y_{97}, y_{98}, y_{99}, y_{100}]$$

```
> eqns_fun := unapply(Vector([seq(simplify(subs(y[0]=ya,y[n]=yb,
  eqns[i]),sqrt,symbolic),i=1..n-1)]),
  seq(y[k],k=1..n-1)):
  vars_reduced := [seq(y[i],i=1..n-1)]:
> J := unapply(Matrix(simplify(Jacobian(eqns_fun(seq(y[k],k=1..n
  -1)),[seq(y[k],k=1..n-1)]),sqrt,symbolic)),seq(y[k],k=1..n-1)):
> Newton_update := Z -> Z-LinearSolve(J(seq(Z[k],k=1..n-1)),
  eqns_fun(seq(Z[k],k=1..n-1)));
```

Newton_update := *Z* → *VectorCalculus*:-

+(*Z*

, *VectorCalculus*:--(*LinearAlgebra*:-

$LinearSolve\left(J\left(seq\left(Z_k, k=1..VectorCalculus{:-}{+}(n, VectorCalculus{:-}{-}(1))\right)\right)\right),$

$eqns\_fun\left(seq\left(Z_k, k=1..VectorCalculus{:-}{+}(n, VectorCalculus{:-}{-}(1))\right)\right)\right)\right)\right)$

```
> Z0 := <seq(1,k=1..n-1)>;
```

$$Z0 := \begin{bmatrix} \textit{1 .. 99 } Vector_{column} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix}$$

```
> Z1 := evalf(Newton_update(Z0));
```

$$Z1 := \begin{bmatrix} \textit{1 .. 99 } Vector_{column} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix}$$

```
> Z2 := evalf(Newton_update(Z1));
```

$$Z2 := \begin{bmatrix} \textit{1 .. 99 } Vector_{column} \\ \textit{Data Type: } float_8 \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix}$$

```
> Z3 := evalf(Newton_update(Z2));
```

$$Z3 := \begin{bmatrix} \textit{1 .. 99 } Vector_{column} \\ \textit{Data Type: } float_8 \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix}$$

```
> yyy := [1,seq(Z3[k],k=1..n-1),1]:
  xxx := [seq(k/n,k=0..n)]:
> plot([seq([xxx[k],yyy[k]],k=1..nops(yyy))]);
```


1,0
0,975
0,95
0,925
0,9
0,875
0,85
0,0
0,1
0,2
0,3
0,4
0,5
0,6
0,7
0,8
0,9
1,0

>